

[Kerry Lee](#) posted on Tuesday, November 30, 2010 - 12:41 am  
Dear Dr Muthen,

I have been going through the posts on the testing of nested CFA models and am getting a bit confused.

The scenario that I am concerned with involves testing whether a number of observed variables fit a 1, 2, or 3 factor CFA model (similar to the example given in [.../9/1611.html](#)). In a reply to that post, it was stated that forcing the correlation between two factors to 1 violates a key assumption in testing nested model using chi-squared.

In [...9/344.html](#), similar advice was given: a 2 factor model in such scenarios is not nested under a 3 factor model.

Yet, in post [.../9/5167.html](#), an elegant solution seems to be available using the MODEL TEST command. Rather than fix the correlation between two factors to 1, a Wald test can be used to test whether the the correlation between two factors differ significantly from 1.

I guess if the Wald test is not significant, one can then infer that the correlation in question does not differ from zero and adopt the model with fewer factors.

Because the Wald test is applied to only one model, the question of "nestedness" does not technically arise. Can I please confirm whether this line of reasoning is correct and that using the Wald test in this way allows for the evaluation of a 2 versus 3 factor model?

Sincerely,  
Kerry.

---

⚡  
Bengt O. Muthen posted on Wednesday, December 01, 2010 - 10:20 am  
Testing on the border of the admissible parameter space (corr=1 being one example, v=0 another) could affect both LRT chi-square testing and Wald testing (z testing in a univ setting). But the question remains how well or poorly it works in practice. Next, I give input for 2 simulation studies I did to explore testing 1 factor vs 2 CFA factors. You can run it and look at the good results for the Wald test. And then generalize to your setting.

---

⚡  
Bengt O. Muthen posted on Wednesday, December 01, 2010 - 10:30 am  
Here is Monte Carlo input for generating data from a 1-factor model and analyzing with a 2-factor CFA. The question is what the empirical p-value is for the Wald test: How often do we reject that the 2 factors correlate one? The p-value is 0.06 which is close to the correct 0.05, that is, we reject that the factor correlation is one at the correct low Type I error level. In other words, we correctly choose 1 factor. The p-value 0.06 is also seen in the last power column for the New parameter diff.

title: 1f 2f  
montecarlo:  
names = y1-y8;

```

nobs = 200;
nreps = 500;
model population:
f by y1-y8*.8;
f*1;
y1-y8*.36;
model:
f1 by y1-y4*.8;
f2 by y5-y8*.8;
f1-f2@1;
f1 with f2*1 (corr);
y1-y8*.36;
model constraint:
new(diff*0);
diff = 1-corr;
model test:
0 = 1-corr;

```

---



Bengt O. Muthen posted on Wednesday, December 01, 2010 - 10:34 am  
Here is Monte Carlo input for generating data from a 2-factor CFA and analyzing with the same model. Here the empirical 5% p-value and power estimate is 0.74 at n=200, which approaches good power to reject the false hypothesis that the factors correlate 1. Note that this is the case despite the true correlation being high, 0.95.

```

title: 2f 2f
montecarlo:
names = y1-y8;
nobs = 200;
nreps = 500;
model population:
f1 by y1-y4*.8;
f2 by y5-y8*.8;
f1-f2*1;
f1 with f2*.95;
y1-y8*.36;
model:
f1 by y1-y4*.8;
f2 by y5-y8*.8;
f1-f2@1;
f1 with f2*.95 (corr);
y1-y8*.36;
model constraint:
new(diff*.05);
diff = 1-corr;
model test:
0 = 1-corr;

```